Maple-assisted proof of empirical formula for A262912

Robert Israel

Aug 12 2019

Let the "state" of a 4-column array of 0..1 be the values mod 7 of the columns considered as binary numbers. Thus there are $7^4 = 2401$ possible states to consider. The possible rows correspond to binary numbers of up to 4 digits that are divisible by 3, namely $0000_2 = 0$, $0011_2 = 3$, ... $1111_2 = 15$.

. Let T be the 2401×2401 matrix such that T_{ij} is the number of possible rows r such that an array in state number i atop row r will produce an array in state number j. Thus if the first array's state is (w, x, y, z) and the row is (s, t, u, v), the new array has state $(2\,w+s, 2\ x+t, 2\,y+u, 2\,z+v)$ (with operations done mod 7). Each row of T has six nonzero entries. The following code produces the matrix T.

```
> States:= [seq(seq(seq(seq([w,x,y,z],z=0..6),y=0..6),x=0..6),w=0.
  .6)]:
  T:= Matrix(2401,2401,storage=sparse):
  for i from 1 to 2401 do
     for y in [[0,0,0,0],[0,0,1,1],[0,1,1,0],[1,0,0,1],[1,1,0,0],
  [1,1,1,1]] do
        z:= 2*States[i]+y mod 7;
        j:= 7^3*z[1]+7^2*z[2]+7*z[3]+z[4]+ 1;
        T[i,j]:= 1;
  od od:
```

Considering an array of 0 rows as being in state $(0, 0, 0, 0)$, which is the first in our enumeration, $a(n)$ should be $e_1^T T^{n+1} e_1$ where e_1 is the unit vector with 1 in the first position. To confirm, we compute the first few entries. For future use, $U_j = T^j e_1$.

```
> U[0]:= Vector([1,0$2400]):
  for j from 1 to 30 do U[j]:= T . U[j-1] od:
```

```
> seq(U[n+1][1],n=1..27);
```

1, 6, 15, 53, 318, 1207, 5797, 34782, 189135, 1089701, 6538206, 38547751, 229660021, 1377960126, 8242589055, 49395098933, 296370593598, 1777250964247, 10661181588037, 63967089528222, 383764138693935, 2302493636842181, 13814961821053086, 82888234267090951, 497325775376734741, 2983954652260408446, 17903665903505468895 **(1)**

Now the empirical formula is

```
> Emp:= a(n) = 8*a(n-1) -16*a(n-2) +104*a(n-3) -640*a(n-4) +1280*a
  (n-5) -3830*a(n-6) +15280*a(n-7) -30560*a(n-8) +58240*a(n-9)
  -99200*a(n-10) +198400*a(n-11) -283209*a(n-12) -115128*a(n-13)
  +230256*a(n-14) -345384*a(n-15);
```

$$Emp := a(n) = 8\,a(n-1) - 16\,a(n-2) + 104\,a(n-3) - 640\,a(n-4) + 1280\,a(n-5) - 3830\,a(n-6) + 15280\,a(n-7) - 30560\,a(n-8) + 58240\,a(n-9) - 99200\,a(n-10) + 198400\,a(n-11) - 283209\,a(n-12) - 115128\,a(n-13) + 230256\,a(n-14) - 345384\,a(n-15) \quad \textbf{(2)}$$

This corresponds to saying $e_1^T\,P(T)\,T^{n+1} e_1 = 0$ where P is the following polynomial.

```
> P:= t^15 - add(coeff(rhs(Emp),a(n-j))*t^(15-j),j=0..15);
```

$$P := t^{15} - 8\,t^{14} + 16\,t^{13} - 104\,t^{12} + 640\,t^{11} - 1280\,t^{10} + 3830\,t^{9} - 15280\,t^{8} + 30560\,t^{7} - 58240\,t^{6} + 99200\,t^{5} - 198400\,t^{4} + 283209\,t^{3} + 115128\,t^{2} - 230256\,t + 345384 \qquad (3)$$

We compute $v = P(T)\,e_1$ using the previously computed values U_j.

```
> v:= add(coeff(P,t,j)*U[j],j=0..15):
```

It would be convenient if $P(T)\,e_1 = 0$, but that is not the case. However, it is true that $P(T)\,e_1$ is an eigenvector of T for eigenvalue -2.

```
> LinearAlgebra:-Equal(T.v + 2*v,Vector(2401,0));
```

$$\mathit{true} \qquad (4)$$

Moreover, $e_1^T v = 0$.

```
> v[1];
```

$$0 \qquad (5)$$

Thus for any positive integer k we have $e_1^T\,P(T)\,T^{n+1}\,e_1 = e_1^T\;T^{n+1}v = (-2)^{n+1}\,e_1^T\;v = 0$. This completes the proof.